Raabo.

A/S REGNECENTRALEN
SCANDINAVIAN INFORMATION PROCESSING SYSTEMS

SYSTEM LIBRARY

RCSL NO: 55-D58

TYPE : Algol 5 Procedure

AUTHOR : J. Winther

EDITION: September 1969

RC 4000 SOFTWARE

MATHEMATICAL PROCEDURE LIBRARY

gamma

ABSTRACT

The real procedure gamma(z) approximates the gamma function of the real or positive integer argument z.

INFORMATION DEPARTMENT

DK-2500 VALBY · BJERREGAARDSVEJ 5 · PHONE: (01) 46 08 88 · TELEX: 6464 rcinf dk · CABLES: INFOCENTRALEN

Gamma function, gamma(z)

1. Function and parameters.

gamma(z) approximates the gamma function in the range -301<z<301.

Procedure heading:

```
    real procedure gamma(z);
    value z; real z;
```

Procedure identifier:

gamma : (real)
approximated function of an argument not resulting in under- or overflow, in which case gamma is undefined.

Call parameter:

z : (real or integer)
argument, values equal to nonpositive integers and values exceeding the range above will give floating point under- or overflow.

2. Method.

The value of gamma(2+x) is approximated in the range 0<=x<=1 by a rational function given as approximation 5231 in (1) with the numerator degree 6 and denominator degree 3.

For arguments outside the basic range 2<=z<=3, successive multiplications or divisions are performed according to the recurrence formula:

gamma(z+1)=z×gamma(z)

The value of the argument is not controlled in any way.

3. Accuracy, Time- and Storage Requirement.

3.1 Accuracy.

The error estimates given below assume the argument to be exactly represented.

Outside the range -2<z<7 the increment in gamma(z) caused by an increment of one unit in the last binary place of z will be greater than the computational error of the procedure in any case.

The error estimates are given as functions of:
u:= abs(entier(z-2))+6

max rel error : u×2.9₁₀-11
safe upper bound for the relative error of gamma(z).

rel mean error: sqrt(u)×1.2₁₀-11
relative mean error (standard error) of gamma(z) assuming a random distribution of rounding errors and mantissas of floating point numbers.
The probability of a relative error greater than 3×(rel mean error) is less than 0.01.

3.2 Time Requirement.

Approximate cpu-time: z>=2: 720 + entier(z-2)×81 usec
z< 2: 720 + entier(z-2)×72 usec

3.3 Storage Requirement:

Codelength: 1 segment.

Typographical length: 47 lines incl. last comment.

4. Test and discussion.

The procedure has been compared with a double precision procedure and with values of the gamma function given in (2). The results are in accordance with the theoretical estimate of the mean error.

Simple testprogram with data and output:

```
begin

comment: here the procedure is copied (without the first
         3 lines) unless it is already translated as an
         external;

  real z, g;
    write(out,<:<12>
      z                            gamma(z)<10>:>);
AGAIN:
    overflows:=underflows:=0;
    read(in,z);
    write(out,<:<10>:>,<<-d.dd ddd ddd ddd'-ddd>, z);
    if z>1000 then goto FINISH;
    g:=gamma(z);
    if overflows>0 then
      write(out, false add 32, 15, <:overflow:>)
    else if underflows>0 then
           write(out, false add 32, 15, <:underflow:>)
         else write(out,
                 <<        -d.dd ddd ddd ddd'-ddd>, g);
    goto AGAIN;
FINISH:
end;
```

data: 0.5, 1, 10, 301, 302,-0.5,-300.9,-301.9,1001

output:

```
            z                             gamma(z)

 5.00 000 000 000'  -1          1.77 245 385 088'   0
 1.00 000 000 000'   0          1.00 000 000 000'   0
 1.00 000 000 000'   1          3.62 880 000 000'   5
 3.01 000 000 000'   2          3.06 057 512 208' 614
 3.02 000 000 000'   2                overflow
-5.00 000 000 000'  -1         -3.54 490 770 176'   0
-3.00 900 000 016'   2         -1.95 307 772 968'-616
-3.01 900 000 016'   2                underflow
 1.00 100 000 000'   3
end
```

5. References.

(1) J.F.Hart and oth.:
Computer Approximations,
John Wiley and Sons, 1968, p.130-136

(2) M.Abramowitz and I.H.Stegun:
Handbook of Math. Functions,
National Bureau of Standards, 1965, p.253-275.

6. Algorithm.

```
gamma = set 1
gamma = algol
external

  real procedure gamma(z);
  value z; real z;
  begin
    real h;
    h:=1.0;
    if z>2.0 then
      begin
        for z:=z-1.0 step -1.0 until 2.0 do h:=h×z;
        z:=z-1.0
      end
    else if z<1.0 then
           begin
             for z:=z step 1.0 until 0.0 do h:=h/z;
             h:=h/z/(z+1.0)
           end
         else begin  h:=h/z; z:=z-1.0  end;
    gamma:=((((((+.039 301 346 419)×z+.142 928 007 949)×z
                 +1.09 850 630 453)×z+3.36 954 359 131)×z
                 +12.8 021 698 112)×z+22.9 680 800 836)×z
                 +43.9 410 209 189)/
              ((((+1.00 000 000 000)×z-7.15 075 063 299)×z
                 +4.39 050 474 596)×z+43.9 410 209 191)×h
  end gamma;
```

comment:

gamma(z) approximates the gamma function in the range -301<z<301.

Procedure identifier:

gamma : (real) approximated function of an argument not resulting in under- or overflow, in which case gamma is undefined.

Call parameter:

z : (real or integer) argument, values equal to nonpositive integers and values exceeding the range above will give floating point under- or overflow.